
NSY116 - Multimédia et interaction humain-machine

ED - Son numérique

Question 1 : Analyse harmonique

Soit le signal en dent de scie déterminé par 12 échantillons :

Ecrire un programme Processing pour calculer numériquement l'amplitude des harmoniques
(on se préoccupe pas de la période du signal)

Question 2 : Synthèse additive

Avec un logiciel de tracé de courbe (gnuplot, etc.), visualiser le résultat obtenu par addition
successive des harmoniques pour le signal de la question 1

Question 3 : FFT

Comparer le résultat obtenu question 1 avec une analyse FFT effectuée avec Processing et la
librairie ESS

Expliquer les différences

Question 4 : Logiciels

Faire la liste de tous les logiciels de traitement et synthèse sonore que vous connaissez et
classez les selon leur mode d'interaction avec l'utilisateur. Idem pour les standard et les
formats de fichiers.



CORRECTION

Question 1

/// calcul des coef. transformee de Fourier

float[] u = {0,0.262,0.524,0.786,1.047,1.309,0,-1.309,-1.047,-0.786,-0.524,-0.262};
int  N = 12;
int R = 6;
float a,b,c;

a = 0;
for (int k=0;k<N;k++) {
 a += u[k];
}
a *= 1.0/N;
println("a0 : "+a);

for (int i=1;i<R;i++) {
 a = 0;
 for (int k=0;k<N;k++){
  a += u[k]*cos(2.0*PI*i*k/N);
 }
 a *= 2.0/N;
 b = 0;
 for (int k=0;k<N;k++){
  b += u[k]*sin(2.0*PI*i*k/N);
 }
 b *= 2.0/N;
 c = sqrt(a*a + b*b);
 println( i+" : "+a+" "+b+" "+c);
}

On obtient :

a0 : -7.4505806E-9
1 : -1.291434E-7     0.97734207        0.97734207
2 : 1.1175871E-7     -0.45321998        0.45321998
3 : -2.9585294E-7       0.2616666        0.2616666
4 : 2.8808913E-7         -0.15126565      0.15126565
5 : 1.1920929E-7         0.07032463       0.07032463

donc le signal est représenté par la série :
F(t) = 0 + 0*cos(t) + 0*cos(2t) + … + 0*cos(5t)
          + 0.977*sin(t) - 0.4532*sin(2t) + 0.2617*sin(3t) - 0.1513*sin(4t) + 0.0703*sin(5t)



Question 2   Exemple de tracé avec Grapher sur MacOs

 la fondamentale : 0.977*sin(t)

 deux harmoniques 0.977*sin(t) - 0.4532*sin(2t)

 trois

 quatre

 cinq :
il reste des ondulations



Question 3

On génère dans un premier temps un fichier son de signal en dent de scie (ex. ici du logiciel
Audacity). La fréquence du signal choisie est 1000 Hz

On utilise ensuite le prog. processing ci-dessous, adapté des exemples de la librairie de
traitement sonore ESS :

// analyse FFT minimale

import krister.Ess.*;

int bufferSize;
int rectwid;

FFT myFFT;
AudioInput myInput;

void setup () {
  size(700,200);
  // start up Ess
  Ess.start(this);
  // set up our AudioInput
  bufferSize=512;
  myInput=new AudioInput(bufferSize); // par def. Fe = 44100 Hz
  // set up our FFT
  myFFT=new FFT(bufferSize*2);

  // dimension des barres du spectre
  rectwid = int(width/bufferSize);

  frameRate(25);
  myInput.start();
}

void draw() {
  //background(255,255,255);
  noStroke();
  fill(0,0,255);

  // draw the spectrum



  for (int i=0; i<bufferSize; i++) {
    rect(i*rectwid,0,rectwid,myFFT.spectrum[i]*500);
  }

  // trace les frequences tous les 1000Hz
  stroke(255,0,0);
  for (int i=1; i<=22; i++) {
    line(i*width/22,0,i*width/22,height);
  }
}

public void audioInputData(AudioInput theInput) {
  myFFT.getSpectrum(myInput);
}

void mousePressed() {
 background(255,255,255);
}

Qui nous affiche le sonagramme suivant :

On retrouve clairement des harmoniques, mais pas vraiment "en face" des graduations de
fréquence prévues. Il faut aussi expliquer les "ébarbures" autour des harmoniques. Si on
analyse un signal sinusoïdal pur à 1000 Hz, on obtient le même genre de chose :

La grande différence entre une analyse à la volée comme ici et celle de la question 1 est qu'il
n'y a pas de synchronisme entre le processus d'échantillonage et la fréquence d'origine du
signal. Ce problème est inhérent à la transformée de Fourier discrète.

Remarque : les amplitudes relatives des harmoniques ne sont pas non plus bien respectées (la
2eme est plus forte que la fondamentale), mais le signal triangulaire généré avec Audacity
n'est pas exactement celui analysé aux questions précédentes (front montant plus brutal).


